
CSCI 5330
Digital CMOS VLSI Design

Instructor: Saraju P. Mohanty, Ph. D.

Lecture 11 : Circuit Simulation

NOTE: The figures, text etc included in slides are borrowed 
from various books, websites, authors pages, and other 
sources for academic purpose only. The instructor does 
not claim any originality.

Outline

• Introduction to SPICE
• DC Analysis
• Transient Analysis
• Subcircuits
• Optimization
• Power Measurement
• Logical Effort Characterization



Dept. of Comp. Sc. & Eng. CSCI 5330 : Digital CMOS VLSI Design 3

Simulators at Different Levels

• Simulators at different levels of abstraction : 
– Process
– Circuit
– Logic

• Process Simulator: Time and temperature effect 
on physical and electrical characteristics.

• Circuit Simulator: Use device model and netlist  
to predict current, voltage, and performance.

• Logic Simulator: Functional prediction of digital 
circuits for logical correctness.

Introduction to SPICE

• Simulation Program with Integrated Circuit 
Emphasis
– Developed in 1970’s at Berkeley
– Many commercial versions are available
– HSPICE is a robust industry standard

• Has many enhancements that we will use

• The circuits elements are called cards, and 
complete description is called a SPICE deck.

• The file contains a netlist consisting of 
components and nodes, simulation options, 
analysis options, and device models.

Writing Spice Decks

• Writing a SPICE deck is like writing a good 
program
– Plan: sketch schematic on paper or in editor

• Modify existing decks whenever possible

– Code: strive for clarity
• Start with name, email, date, purpose
• Generously comment

– Test:
• Predict what results should be
• Compare with actual
• Garbage In, Garbage Out!

SPICE : RC Circuit Example
* rc.sp
* David_Harris@hmc.edu 2/2/03
* Find the response of RC circuit to rising input
*------------------------------------------------
* Parameters and models
*------------------------------------------------
.option post
*------------------------------------------------
* Simulation netlist
*------------------------------------------------
Vin in gnd pwl 0ps 0 100ps 0 150ps 1.8 800ps 1.8
R1 in out 2k
C1 out gnd 100f
*------------------------------------------------
* Stimulus
*------------------------------------------------
.tran 20ps 800ps
.plot v(in) v(out)
.end

R1 = 2KΩ

C1 =
100fF

Vin
+

Vout
-

SPICE : RC Circuit Example …

Result (Textual)

SPICE : RC Circuit Example …

 t(s)
0.0 100p 200p 300p 400p 500p 600p 700p 800p 900p

0.0

0.5

1.0

1.5

2.0 v(in)

v(out)

Result (Graphical)

SPICE Sources

• DC Source
Vdd vdd gnd 2.5

• Piecewise Linear Source
Vin in gnd pwl 0ps 0 100ps 0 150ps 1.8 800ps 1.8

• Pulsed Source
Vck clk gnd PULSE 0 1.8 0ps 100ps 100ps 300ps 800ps

PULSE v1 v2 td tr tf pw per

v1

v2

td tr tfpw

per

SPICE Elements
Letter Element
R Resistor
C Capacitor
L Inductor
K Mutual Inductor
V Independent voltage source
I Independent current source
M MOSFET
D Diode
Q Bipolar transistor
W Lossy transmission line
X Subcircuit
E Voltage-controlled voltage source
G Voltage-controlled current source
H Current-controlled voltage source
F Current-controlled current source

SPICE Units

109gigag
106megax
103kilok
10-3milim
10-6microu
10-9nanon
10-12picop
10-15femptof
10-18attoa
MagnitudeUnitLetter

Ex: 100 femptofarad capacitor = 100fF, 100f, 100e-15

Transistor DC Analysis
* mosiv.sp
*------------------------------------------------
* Parameters and models
*------------------------------------------------
.include '../models/tsmc180/models.sp'
.temp 70
.option post
*------------------------------------------------
* Simulation netlist
*------------------------------------------------
*nmos
Vgs g gnd 0
Vds d gnd 0
M1 d g gnd gnd NMOS W=0.36u

L=0.18u
*------------------------------------------------
* Stimulus
*------------------------------------------------
.dc Vds 0 1.8 0.05 SWEEP Vgs 0 1.8 0.3
.end

Vgs Vds

Ids

4/2

Transistor DC Analysis
• nMOS I-V

– Vgs dependence
– Saturation

Vds

0.0 0.3 0.6 0.9 1.2 1.5 1.8

Ids
(µA)

0

50

100

150

200

250

Vgs = 1.8

Vgs = 1.5

Vgs = 1.2

Vgs = 0.9

Vgs = 0.6Vgs Vds

Ids

4/2

I-V characteristics

MOSFET Elements

M element for MOSFET

Mname drain gate source body type
+ W=<width> L=<length>
+ AS=<area source> AD = <area drain> 
+ PS=<perimeter source> PD=<perimeter drain>

Inverter Transient Analysis
* inv.sp
* Parameters and models
*------------------------------------------------
.param SUPPLY=1.8
.option scale=90n
.include '../models/tsmc180/models.sp'
.temp 70
.option post
* Simulation netlist
*------------------------------------------------
Vdd vdd gnd 'SUPPLY'
Vin a gnd PULSE 0 'SUPPLY' 50ps 0ps 0ps 100ps 200ps
M1 y a gnd gnd NMOS W=4 L=2 
+ AS=20 PS=18 AD=20 PD=18
M2 y a vdd vdd PMOS W=8 L=2
+ AS=40 PS=26 AD=40 PD=26
* Stimulus
*------------------------------------------------
.tran 1ps 200ps
.end

a y

4/2

8/2

Inverter Transient Analysis ….

 (V)

0.0

1.0

 t(s)
0.0 50p 100p 150p 200p

v(a)

v(y)

tpdr = 15pstpdf = 12ps

tf = 10ps

tr = 16ps

0.36

1.44

1.8

• Unloaded inverter
– Overshoot
– Very fast 

edges

Subcircuits

• Declare common elements as subcircuits

• Ex: Fanout-of-4 Inverter Delay
– Reuse inv
– Shaping
– Loading

.subckt inv a y N=4 P=8
M1 y a gnd gnd NMOS W='N' L=2 
+ AS='N*5' PS='2*N+10' AD='N*5' PD='2*N+10'
M2 y a vdd vdd PMOS W='P' L=2
+ AS='P*5' PS='2*P+10' AD='P*5' PD='2*P+10'
.ends

a b c d eX1 X2 X3 X4
1

2

4

8

16

32

64

128 fX5
256

512

Shape input

Device
Under
Test Load

Load on
Load

FO4 Inverter Delay
* fo4.sp
* Parameters and models
*----------------------------------------------------------------------
.param SUPPLY=1.8
.param H=4
.option scale=90n
.include '../models/tsmc180/models.sp'
.temp 70
.option post
* Subcircuits
*----------------------------------------------------------------------
.global vdd gnd
.include '../lib/inv.sp'
* Simulation netlist
*----------------------------------------------------------------------
Vdd vdd gnd 'SUPPLY'
Vin a gnd PULSE 0 'SUPPLY' 0ps 100ps 100ps 500ps 1000ps
X1 a b inv * shape input waveform
X2 b c inv M='H'    * reshape input waveform
.end

FO4 Inverter Delay ….
X3 c d inv M='H**2' * device under test
X4 d e inv M='H**3' * load
x5 e f inv M='H**4' * load on load
* Stimulus
*----------------------------------------------------------------------
.tran 1ps 1000ps
.measure tpdr * rising prop delay
+     TRIG v(c) VAL='SUPPLY/2' FALL=1 
+     TARG v(d)  VAL='SUPPLY/2' RISE=1
.measure tpdf * falling prop delay
+     TRIG v(c)  VAL='SUPPLY/2' RISE=1
+     TARG v(d)  VAL='SUPPLY/2' FALL=1 
.measure tpd param='(tpdr+tpdf)/2' * average prop delay
.measure trise * rise time
+ TRIG v(d) VAL='0.2*SUPPLY' RISE=1
+ TARG v(d) VAL='0.8*SUPPLY' RISE=1
.measure tfall * fall time
+ TRIG v(d) VAL='0.8*SUPPLY' FALL=1
+ TARG v(d) VAL='0.2*SUPPLY' FALL=1
.end 

FO4 Inverter Delay ….

 (V)

0.0

0.5

1.0

1.5

2.0

 t(s)
0.0 200p 400p 600p 800p 1n

a

b

c

d

e

f
tpdf = 66ps tpdr = 83ps

Optimization

• HSPICE can automatically adjust parameters
– Seek value that optimizes some measurement

• Example: Best P/N ratio
– We’ve assumed 2:1 gives equal rise/fall delays
– But we see rise is actually slower than fall
– What P/N ratio gives equal delays?

• Strategies
– (1) run a bunch of sims with different P size
– (2) let HSPICE optimizer do it for us

P/N Optimization
* fo4opt.sp
* Parameters and models
*----------------------------------------------------------------------
.param SUPPLY=1.8
.option scale=90n
.include '../models/tsmc180/models.sp'
.temp 70
.option post
* Subcircuits
*----------------------------------------------------------------------
.global vdd gnd
.include '../lib/inv.sp'
* Simulation netlist
*----------------------------------------------------------------------
Vdd vdd gnd 'SUPPLY'
Vin a gnd PULSE 0 'SUPPLY' 0ps 100ps 100ps 500ps 
1000ps
X1 a b inv P='P1' * shape input 
waveform
X2 b c inv P='P1' M=4 * reshape input
X3 c d inv P='P1' M=16 * device under test

P/N Optimization ….
X4 d e inv P='P1' M=64 * load
X5 e f inv P='P1' M=256 * load on load
* Optimization setup
*----------------------------------------------------------------------
.param P1=optrange(8,4,16) * search from 4 to 16, guess 8
.model optmod opt itropt=30 * maximum of 30 iterations
.measure bestratio param='P1/4' * compute best P/N 
ratio
* Stimulus
*----------------------------------------------------------------------
.tran 1ps 1000ps SWEEP OPTIMIZE=optrange RESULTS=diff 
MODEL=optmod
.measure tpdr * rising propagation delay
+       TRIG v(c) VAL='SUPPLY/2' FALL=1 
+       TARG v(d)  VAL='SUPPLY/2' RISE=1
.measure tpdf * falling propagation delay
+       TRIG v(c)  VAL='SUPPLY/2' RISE=1
+       TARG v(d)  VAL='SUPPLY/2' FALL=1 
.measure tpd param='(tpdr+tpdf)/2' goal=0 * average prop delay
.measure diff param='tpdr-tpdf' goal = 0 * diff between delays
.end 

P/N Optimization ….

• P/N ratio for equal delay is 3.6:1
– tpd = tpdr = tpdf = 84 ps (slower than 2:1 ratio)
– Big pMOS transistors waste power too
– Seldom design for exactly equal delays

• What ratio gives lowest average delay?

.tran 1ps 1000ps SWEEP OPTIMIZE=optrange RESULTS=tpd
MODEL=optmod

– P/N ratio of 1.4:1
– tpdr = 87 ps, tpdf = 59 ps, tpd = 73 ps



Dept. of Comp. Sc. & Eng. CSCI 5330 : Digital CMOS VLSI Design 25

Device Models

• SPICE uses wide variety of transistor models 
with different trade-offs between complexity and 
accuracy.

• Different models:
– Level 1 (simple)
– Level 2 (simple)
– Level 3 (simple)
– BSIM 1 (Berkley Short-Channel IGFET Model)
– BSIM 2 (very elaborate)
– BSIM 3 (very elaborate)
– BSIM 4 (current, most elaborate) 

Device Models : Level 1

• Closely related to the Shockley model
• Enhanced with channel length modulation and the body 

effect.
• The parameters from the SPICE model are in CAPS.
• The threshold voltage is modulation by the source-to-

body voltage (Vsb) due to body effect is taken into 
account.

• Gate capacitance is calculated from the oxide thickness 
TOX.

Device Models : Level 2 and Level 3

• Account effects of :
– Velocity Saturation
– Mobility Degradation
– Subthrehold Conduction
– Drain-Induced Barrier Lowering

• Still not good enough to mimic the 
characteristics of the modern devices.

Device Models : BSIM

• Features of version 3v3:
– Continuous and differentiable I-V characteristics for 

different regions of operation for good convergence.
– Sensitivity of Vt, etc to length and width
– Vt model include body effect and DIBL.
– Velocity saturation, mobility degradation, etc.
– Multiple gate capacitance
– Diffusion capacitance and resistance models

• Features of version 4:
– Tunneling current
– High-K dielectric

Design Corners

Device Characterization : I-V characteristics

180nm process, Vdd = 1.8V, 
W = 1000nm, 70oCI ds
 

Vds 

I ds
 

Vds

360nm process, Vdd = 1.8V, 
W = 1000nm, 70oC

Device Characterization : Vt

Two methods to determine 
Vt from Ids-Vgs
characteristics:

•Constant Current

•Linear Extrapolation

Constant Current: 
•Vt is defined as the gate 
voltage at a given drain 
current.

•Depends on an arbitrary 
choice of a drain current.

Vgs 

I ds

Device Characterization : Vt

Linear Extrapolation: 

• Extrapolates the gate voltage from 
the point of maximum slope on Ids-
Vgs characteristics.

• Valid for linear region of operation 
only.

I ds
 Vgs 

Device Characterization : Gate Capacitance

• Effective gate capacitance is needed to estimate 
the delay (RC).

Device Characterization : Gate Capacitance

Device Characterization : Parasitic Capacitance

• Diffusion capacitance is voltage dependent

• Effective gate capacitance is obtained as average over switching
transition.

Device Characterization : Effective Resistance

• RC delay model:  Unit transistor has capacitance C, 
parasitic resistance Cd, resistance Rn for NMOS, 
resistance Rp for PMOS.

• As C is known from effective capacitance extraction, Rn
the and Rp values can be know once delays are 
measured.

Circuit Characterization : Power

• HSPICE can measure power
– Instantaneous P(t)
– Or average P over some interval

.print P(vdd)

.measure pwr AVG P(vdd) FROM=0ns TO=10ns

• Power in single gate
– Connect to separate VDD supply
– Be careful about input power

Circuit Characterization : Monte Carlo Simulation

• Used to find the effect of random variations on a circuit.
• Consists of running simulation repeatedly with different 

randomly chosen parameters.
• Transistor models must include the offset parameters.

Circuit Characterization : Logical Effort

• Logical effort can be measured from simulation
– As with FO4 inverter, shape input, load output

X1 X2 X3 X4 X5

a b c d e
fM=1 M=h M=h2

M=h3
M=h4

Shape input

Device
Under
Test Load

Load on
Load

Circuit Characterization : Logical Effort

• Plot tpd vs. h
– Normalize by τ
– y-intercept is parasitic delay
– Slope is logical effort

• Delay fits straight line
very well in any process
as long as input slope is
consistent

